# Maintainer: Maxime Gauduin <alucryd@archlinux.org>

pkgname=wingpanel-indicator-session
pkgver=2.3.1
pkgrel=2
pkgdesc='Session indicator for Wingpanel'
arch=(loong64 x86_64)
url=https://github.com/elementary/wingpanel-indicator-session
license=(GPL3)
groups=(pantheon)
depends=(
  accountsservice
  gdk-pixbuf2
  glib2
  gtk3
  libgranite.so
  libhandy-1.so
  wingpanel
)
makedepends=(
  git
  gobject-introspection
  meson
  vala
)
_tag=ade43580559fb7dff8c2ae725677c700482d661c
source=(git+https://github.com/elementary/wingpanel-indicator-session.git#tag=${_tag})
sha256sums=(SKIP)

pkgver() {
  cd wingpanel-indicator-session
  git describe --tags
}

build() {
  arch-meson wingpanel-indicator-session build
  meson compile -C build
}

package() {
  DESTDIR="${pkgdir}" meson install -C build
}

# vim: ts=2 sw=2 et:
